Verilog HDL

Lexical Conventions

- Similar to C
- Whitespace
  - Blank spaces, tabs and newlines
  - \b, \t, \n
  - Whitespace ignored except in
    - Strings and token separation
- Comments
  - // - single line;
  - /* … */ - multiple lines

Operators

- Unary,
  - A = ~b;
- Binary
  - a = b&&c;
- Ternary
  - a = b ? c : d;

Number Specifications

- • Sized numbers :

  <size>’<base format><number>

  – 4’b1111; 12’habc

  – <size> - number of bits

- Unsized numbers

  – 23456 // default is decimal

  – ‘hc3

Number Specification

- Negative numbers
  - <size>'<base format><number>
    - -6'd3 // 6-bit negative number stored as two's complement of 3
    - 4'd-2 //illegal specification

Possible Values

- X or Z values
  - 12'h13x; 6'hx; 32'bz;
  - x or z sets 1,3,4 bits in binary, octal and hexadecimal representations respectively
  - Value extension
    - If most significant bit is
      - 0,x,z then 0,x,z respectively
      - 1 then 0
- Data types
  - Value set – 0,1,x, z
  - Signal strength

Readability

- Readability enhancements
  - 12'b1111_1110_0101 //Underscore
- Strings – sequence of ASCII bytes
  - "Hello Verilog Word"; // a string

Identifiers and Keywords

- reg value; // reg – keyword
- input clk; // input – keyword
- More keywords as we progress
- Identifiers made of alphanumeric characters, the underscore (_) and the dollar ($) sign and starts with alphabets or underscore.
- The dollar sign as first character is reserved for system tasks.
  - Ex : $monitor

Nets and Regs

- • Nets
  - Connection between hardware elements
    - wire a; wire b,c; wire d = 1'b0;
    - wand, wor, tri, triand, trior and trireg are other forms of net
- Regs
  - Data storage elements
  - Not equivalent to hardware registers
  - A variable that stores value
  - Unlike a net, it needs no driver
  - Default value for a 'reg' variable is x

Vectors

- Nets and register
- wire [7:0] busA, busB;
- reg [16:0] address;
- [high# : low#] or [low# : high#], but the most significant bit is the left number in the square bracket
- reg [0 : 40] addr1; wire [15 : 12] addr2;

Addressing Vectors

- Address bits are parts of vectors
  - reg [7:0] busA;
    - busA[7]; //stands for 7th bit
    - busA[2:0]; // first 3 LSBs
  - wire [0:15] addr1;
    - addr1[2:1] is illegal;
    - addr1[1:2] is correct;
  - Writing to a vector :
    - busA[2:0] = 3'd6;

Numbers

- Integers
  - datatype storing signed numbers
  - reg stores unsigned numbers
  - Default length is 32-bits
  - integer counter;
  - counter = -1; // stores as 32-bit two's complement number

Real Numbers

- Decimal and Scientific Notations
  - real delta;
    - delta = 4e10;
    - delta = 2.13;
  - integer a;
    - a = delta; /*‘a’ gets the value 2 which is the rounded value of 2.13*/

Time

- Verilog simulation is done with respect to simulation time
- To store simulation time, a special ‘time’ variable is declared in verilog
- The ‘time’ variable is at least 64-bits
- time save_sim_time; save_sim_time = $time;
- Useful for timing measurements
  - Debug

Arrays

- <array_name> [<subscript>]
  - reg bool[31:0]; //compare it with vectors
  - time chk_point [1:100];
  - integer count [0:7];
  - reg [4:0] port_id[0:7];
  - integer matrix_d[4:0][4:0] // illegal
  - count[5]; chk_point[100];
  - How to access say bit 3 of port_id[5];
    - reg [4:0] a;
    - a = port_id[5]; // a[3] is required bit

Arrays vs Vectors

- A vector is a single element that is n-bits wide;
- Arrays are multiple elements that are 1-bit or n-bit wide.

Memories

- reg mem1bit [0:1023];

  // a memory of 1024, 1-bit words
- reg [7:0] membyte [0:1023];

  // a memory of 1024, 8- bit words
- membyte [511];

  //fetches 1 byte word whose address is 511

Parameters

- Imagine this case
  - You design a 4-bit adder with HAs and FAs
  - You need 8-bit adders too
  - Rewrite is a waste of effort
    - Prone to bugs too
- Parameters
  - Method to define constants inside a module
  - For eg. parameter port_id = 5;
  - Can be overriden at module instantiation time

Strings

- Stored in ‘reg’ variables
  - each character takes 8- bits
- reg [8*19:1] string_value;
- string_value = “Hello Verilog world”;
- string_value is defined as a string that can store 19 characters.

System Tasks

- Vendor specific
- Some generic tasks
  - $display
    - similar to printf and used for displaying values of variables and expressions
  - Values can be printed in decimal (%d), binary (%b), string (%s), hex (%h), ASCII character (%c), octal (%o), real in scientific (%e), real in decimal (%f), real in scientific or decimal whichever is shorter (%g).
  - time format (%t), signal strength (%v) and hierarchy name (%m).

System Tasks

- $monitor
  - Monitor a signal when its value changes
  - Only one active monitor list
    - If more than one then, the last one statement will be the active one.
  - Monitoring turned ON by default at start of simulation and can be controlled during the simulation using **$monitoron** and **$monitoroff**

System Tasks

- $stop – stops simulation and puts it into interactive mode
- $finish – terminates the simulation

Continuous Assignments

- assign out = i1 & i2;
- assign addr[15:0] = addr1[15:0]^addr2[15:0];
- assign {cout,sum[3:0]} = a[3:0]+b[3:0]+c_in;
- wire out; assign out = in1 & in2;
- is equivalent to
- wire out = in1 & in2; //Implicit continuous assignment

Expressions

- Dataflow modeling describes the design in terms of expressions instead of primitive gates.
- Expressions – those that combine operands and operators
- a ^ b; addr1[20:17] + addr2[20:17];
- in1 | in2;

Operands

- Constants, integers, real numbers
- Nets, Registers
- Times
- Bit-select
  - One bit of a vector net or vector reg
- Part-select
  - Selected bits of vector net or vector reg
- Memories

Operators -Types

- Arithmetic
- Logical
- Relational
- Equality
- Bitwise
- Reduction // Not available in software languages
- Shift
- Concatenation
- Replication
- Conditional
- Syntax very similar to C

Arithmetic Operators

- Binary Operators
  * - multiply
  / - division
  + - addition
  - subtraction

- Commercial Verilog gives non ‘x’ values whenever possible

  a = 4’b0x11; b = 4’b1000; a+b = 4’b1x11;

  a = 4’b0x11; b = 4’b1100; a+b = 4’bxx11;

Logical Operators

- logical and (&&), logical or (||), logical not (!).
  - They evaluate to a 1-bit value: 0 (false) 1 (true) or x (ambiguous)
  - If an operand is not equal to zero, it is a logical 1 and if it is equal to zero, it is a logical 0. If any operand bit is x or z, then operand is x and treated by simulators as a false condition
  - Logical operators take variables or expressions as operands.

Logical Operators Examples

- A = 3; B = 0;

  A&&B, A||B evaluates to 0 and 1 resp.

  !A, !B evaluates to 0 and 1 resp.

- A = 2'b0x; B = 2'b10;

  A&&B evaluates to x

- (a==2) && (b == 3) //Expressions

Relational Operators

- Greater-than (>)
- Less-than (<)
- Greater-than-or-equal-to (>=)
- Less-than-or-equal-to (<=)
- Evaluates to 1 or 0, depending on the values of the operands
  - If one of the bits is an ‘x’ or ‘z’, it evaluates to ‘x’

Relational Operators (Example)

- //A = 4, B = 3
- //X = 4'b1010, Y = 4'b1101, Z = 4'b1xxx
- A <= B //returns 0
- A > B //returns 1
- Y >= X //returns 1
- Y < Z //returns x

Equality Operators

- Logical equality (==), logical inequality (!=) :

  if one of the bits is ‘x’ or ‘z’, they output ‘x’ else returns ‘0’ or ‘1’

- Case equality (===), case inequality (!==) :

  compares both operands bit by bit and compare all bits including ‘x’ and ‘z’. Returns only ‘0’ or ‘1’

Equality Operators Examples

- //A = 4;B = 3;X = 4'b1010;Y = 4'b1101
- //Z = 4'b1xxz;M = 4'b1xxz;N = 4'b1xxx
- A == B // result is 0
- X != Y //result is 1
- X == Z // result is x
- Z === M //result is 1
- Z === N //result is 0
- M !== N //result is 1

Bitwise operators

- negation (~), and (&), or (|), xor (^), xnor(^~, ~^).
- ‘z’ is treated as ‘x’ in the bitwise operations

and	0	1	x	z
0	0	0	0	0
1	0	1	x	x
x	0	x	x	x
z	0	x	x	x

or	0	1	x	z
0	0	1	x	x
1	1	1	1	1
x	x	1	x	x
z	x	1	x	x

buf	in	out	not	in	out
	0	0		0	1
	1	1		1	0
	x	x		x	x
	z	x		z	x

xor	0	1	x	z
0	0	1	x	x
1	1	0	x	x
x	x	x	x	x
z	x	x	x	x

Bitwise Operators Examples

- //X = 4'b1010;Y = 4'b1101;Z = 4'b10x1
- ~X // result is 4'b0101
- X & Y // result is 4'b1000
- X | Y // result is 4'b1111
- X ^ Y // result is 4'b0111
- X ^~ Y // result is 4'b1000
- X & Z // result is 4'b10x0

Point to Note

- We distinguish between bitwise operators and logical operators
- //X = 4’b1010;Y = 4’b0000
- X | Y // result is 4’b1010
- X || Y // result is 1

Reduction Operators

- Reduction operators perform a bitwise operation on a single vector operand and yield a 1-bit result.
- Reduction operators work bit by bit from right to left.

Reduction Operators Examples

- and (&), nand (~&), or (|), nor (~|), xor (^); and xnor (~^,^~)
- This is a UNARY operation on vectors
- Let X = 4'b1010
- &X is 1'b0 //0 & 1 & 0 & 1 = 1'b0
- |X is 1'b1
- ^X is 1'b0
- A reduction xor or xnor can be used for even or odd parity generation of a vector.

Shift Operators

- Right shift (>>) and left shift (<<)
- //X = 4'b1100
- Y = X >> 1; // Y is 4'b0110
- Y = X << 1; // Y is 4'b1000
- Y = X << 2; // Y is 4'b0000
- Very useful for modeling shift-and-add algorithms for multiplication.

Concatenation Operators

- Denoted by ({,})
- Append multiple ‘sized’ operands. Unsized operands are NOT allowed as size of each operand should be known to compute size of the result
- //A=1’b1;B=2’b00;C=2’b10;D=3’b110;
- Y = {B,C} // Y is 4’b0010
- Y = {A,B,C,D,3’b001} // Y = 11’b10010110001
- Y = {A, B[0],C[1]} // Y is 3’b101

Replication Operators

- Repetitive concatenation of the same number can be represented using a replication constant
- A = 1'b1; B=2'b00;
- Y = { 4{A} }; //Y is 4'b1111
- Y = {4{A},2{B}}; //Y is 8'b11110000

Conditional Operator

- Usage: condition_expr? true_expr : false_expr;
- If condition evaluates to ‘x’, then both expressions are evaluated and compared bit by bit to return for each bit position, an ‘x’ if the bits disagree, else the value of the bit.
- The conditional expression models a 2-to-1 multiplexer
- assign out = control ? in1 : in0;

Compiler Directives

- Usage : `<keyword>
- `define WORD_SIZE 32; compared with #define in C
- `include header.v compared with #include in C.
- `ifdef and `timescale

Modules

- General Structure
  - Module name, port declarations, parameters (optional)
  - Declaration of wire and reg variables
  - Data flow statements (assign)
  - Instantiation of lower level modules
  - always and initial blocks (all behavioral statements)
  - Tasks and Functions
  - endmodule

Ports

- Port declarations – input, output, inout
- Port connection rules
  - there are two ends to a port with respect to a module, one internal and another external
  - Inputs – input ports are to be Nets inside the module and can be reg or net external to the module

Port Connection Rules

- Outputs – Internal reg or net and external should be a net
- Inouts – Both internal and external to be connected to a net
- Width matching – width may not match – only warning issued
- Unconnected ports – again only warning
  - fulladd4(SUM, , A,B, C_IN); the C_OUT is not included

Ports and external Connections

- Connecting by ordered list
- Connecting by name
  - wire C_OUT; wire [3:0] SUM;
  - reg [3:0] A,B; wire C_IN;
  - Fulladd4 f1 (.c_out(C_OUT), .sum(SUM), .a(A), .b(B), .c_in(C_IN));
- Syntax
  - .< name in instantiated module>(name in instantiating module)

Connecting by Name

- Advantages
  - Remembering order of say, 50 ports is difficult
  - Can drop any port during instantiation.
  - Can rearrange the port list of a module without modifying the code that instantiates it.

Verilog Modeling Techniques

- Gate Level Modeling
- Dataflow Modeling (RTL)
- Behavioral Modeling (Algorithmic level)

Gate Level Modeling

- Gate level modeling
  - Logic gate primitives and their instantiation
  - Construct Verilog description from the logic diagram
  - Delay modeling (rise, fall and turn-off)
  - min, max and typ delays in gate level design

Gate Types

- wire OUT, IN1, IN2, IN3;
- and a1(OUT,IN1,IN2);
- //similarly nand, nor, or, xor and xnor
- //extended to more than two inputs
- nand na1_3input(OUT, IN1,IN2,IN3);
- //Name for an instantiation is optional
- and (OUT, IN1, IN2);

Gate Delays

- Rise Delay - 0,x or z → 1
- Fall Delay - 1,x or z → 0
- Turn-off delay – 0, 1 or x → z
- Min of above three for – 0, 1 or z → x

Delay Specification

- No delay specified then default is 0
- If one delay, then used for all transitions
- If two delays, then refers to
  - rise and
  - fall.
  - Turn-off is min of these two.
- If three delays, then refers to
  - rise,
  - fall and
  - turn-off .

Delay Specification Example

- and #(5) a1(out, i1, i2);
- and #(4, 6) a2(out, i1, i2);
- bufif0 #(3, 4, 5) b1 (out,i1,i2);

- assign #10 out = in1 & in2;
- wire #10 out = in1 & in2;
- wire #10 out;
- assign out = in1 & in2;

Delay models

- Three types of delay models used in Verilog
  - Distributed delay model
  - Lumped Delay model
  - Pin-to-pin (path) Delay model

Distributed Delay Model

- Delays that are specified on a *per element* basis

```
module M (out,a,b,c,d);
output out;
input a,b,c,d;
wire e,f;
and #5 a1(e,a,b);
and #7 a2(f,c,d);
and #4 a3(out,e,f);
endmodule
```

```
module M(out,a,b,c,d);
output out;
input a,b,c,d;
wire e,f;
assign #5 e = a & b;
assign #7 f = c & d;
assign #4 out = e & f;
endmodule
```



Lumped delays

- Lumped delays are specified on a *per module* basis.
- Single delay on the output gate of the module – cumulative delays of all paths is lumped at one location.
- They are easy to model compared with distributed delays

Lumped Delay Model Example

```
module M (out,a,b,c,d);
output out;
input a,b,c,d;
wire e,f;
and  a1(e,a,b);
and a2(f,c,d);
and #11 a3(out,e,f);
endmodule
```

```
module M(out,a,b,c,d);
output out;
input a,b,c,d;
wire e,f;
assign  e = a & b;
assign  f = c & d;
assign #11 out = e & f;
endmodule
```



Pin-to-Pin Delays

- Delays are assigned individually to paths from each input to each output.
- Delays can be separately specified for each input/output path

path a-e-out, delay = 9

path b-e-out, delay = 9

path c-f-out, delay = 11

path d-f-out, delay = 11



Behavioral Modeling

- Learning Objectives
  - Use of structured procedures – **always** and **initial**
  - Delay-based, Event-based and Level-sensitive timing controls
  - Conditional statements – **if** and **else**
  - Multiway branching – **case**, **casex** and **casez**
  - Looping Statements – **while**, **for**, **repeat** and **forever**
  - Blocks – **sequential** and **parallel** blocks, **naming** and **disabling** blocks.

The Initial Statement

- An initial block – (initial begin … end) may consist of a group of statements (within begin … end) or one statement
- Each initial block starts at execution time 0
- All initial blocks starts concurrently executing and finishes independent of each other.
- Typically used for initialization, monitoring, waveforms and to execute one-time executable processes.

The always Statement

- Model a block of activity that is repeated continuously in a digital circuit
- Equivalent to an infinite loop
- Stopped only by $finish (power-off) or $stop (interrupt)

Timing Controls

- Delay-Based
  - Regular delay control
  - Intra-assignment delay control
  - Zero delay control

Regular Delay Control

```
parameter latency = 20;
parameter delta = 2;
reg x, y, z, p, q;

initial
begin
        x=0; // no delay control
        #10 y = 1; //delay control with a constant
        #latency z = 0; //delay control with identifier
        #(latency + delta) p = 1; //delay control with expression
        #y x = x + 1; //delay control with identifier
        #(4:5:6) q = 0; //delay with min, typ, and max values
end
```

Intra-assignment delay control

- Assigning delay to the right of the assignment operator
- The intra-assignment delay computes the right-hand-side expression at the current time and defer the assignment of the computed value to the left-hand-side variable.
- Equivalent to regular delays with a temporary variable to store the current value of a right-hand-side expression

Intra-assignment Delay

```
reg x,y,z;
 initial
 begin
            x = 0; z = 0;
            y = #5 x + z;
            //Take value of x and z at the time = 0, evaluate x+z and then
            //wait 5 time units to assign value to y.
            //Any  change to x and z after time = 0 and before 5 will not affect
             the value of y
  end
//The above code is equivalent to
//x = 0; y = 0;                  temp_xz = x + z;
                                 #5 y = temp_xz;
// where, temp_xz is a temporary variable
```

Zero delay control

- Procedural statements inside different always-initial blocks may be evaluated at the same simulation time.
- The order of execution of these statements in different always-initial blocks is nondeterministic and might lead to race conditions.
- Zero delay control ensures that a statement is executed last, after all other statements in that simulation time are executed.
- This can eliminate race conditions, unless there are multiple zero delay statements, in which case again non-determinism is introduced.

Zero delay Control

```
initial
 begin
        x = 0;
        y = 0;
 end
 initial
 begin
        #0 x = 1; //zero delay control
        #0 y = 1;
 end
        //At time = 0, x = 1 and y = 1
```

Event-based timing control

- Regular event control
- Named event control
- Event OR control

Regular Event Control

- @(clock) q = d;
- @(posedge clock) q = d;
- @(negegde clock) q = d;
- q = @(posedge clock) d;

Named event control

```
event received_data
always @(posedge clock)
begin
        if (last_data_packet) -> received_data;
end
always @(received_data)
   data_buf = {data_pkt[0], data_pkt[1]};
```

Event OR control

```
// Sensitivity list
 always @(reset or clock or d)
      // if any one of reset, clock, d changes its value
 begin
        if (reset)
                q = 1'b0;
        else  if (clock)
                q = d;
 end
```

Conditional Statements

- The if-else statement
- if (logical_expression) then <block> else <block>
- <block> = single statement or **begin** <block> **end**
- Nested if-else
  - if <block> else if <block> else if <block

Multiway Branching

- The case statement
- The casex and casez statements

The case statement

- This compares 0,1,x and z of the condition, bit by bit with the different case options.
- If width of condition and a case option are unequal, they are zero filled to match the bid width of the widest of both.

The case Statement

```
//A 4-to-1 Multiplexer
 always @(s1 or s0 or i0 or i1 or i2 or i3)
 case ({s1,s0})
        2'd0: out = i0;
        2'd1: out = i1;
        2'd2: out = i2;
        2'd3: out = i3;
        2'bx0, 2'bx1, 2'bxx,2'b0x,2'b1x: out = 1'bx;
        2'bxz, 2'bzx, 2'bzz: out = 1'bz;
   default: $display("Invalid Control signals");
 endcase
```

The casex and casez statements

- casez does not compare z-values in the condition and case options. All bit positions with z may be represented as ? in that position.
- casex does not compare both x and z-values in the condition and case options

```
casex (encoding)
        4'b1xxx: next_state = 3;
        4'bx1xx: next_state = 2;
        4'bxx1x: next_state = 1;
        4'bxxx1: next_state = 0;
 endcase
//encoding = 4'b10xz will cause
   next_state=3
```

Loops

- while (condition) <block>
- for (count = 0; count < 128; count = count + 1) <block>
- repeat (value) <block>
  - The value should be a constant or variable, and the evaluation of the variable will be done at start of loop and not during execution
- forever loop
  - initial
  - begin
  - clk = 1'b0;
  - forever #10 clock = ~clock;
  - end

Procedural Assignment

- Two types
  - Blocking Assignments
  - Non Blocking Assignments

Blocking Assignments

They are denoted by '=' operator.

In a sequential block it blocks any statement beyond it.

```
reg x,y,z
        x = 0;
        #5 x = 1;     //After 5 units
        #5 y = 1;     //After 10 units
        z = #5 x;     // After 15 units
        #5 x = 0;     // After 20 units
```

In a parallel block, however it does not block.

Non blocking Assignments

- These assignments do not block execution of statements that follow them in a sequential block.
- Denoted by <=

```
    x = 0;
reg_a[2] <= #15 1'b1; // at 15 units
reg_b[15:13] <= #10 {x,y,z} // at 10 units
count <= count + 1; // at 0 units
```

Applications of Non Blocking

- Distinguish between the following
- Type 1 // Race condition – Simulator dependent
  - always @(posedge clock)
  - a=b;
  - always @(posedge clock)
  - b = a;
- Type 2 // ‘a’ swaps with ‘b’ – No Race condition
  - always @(posedge clock)
  - a <= b;
  - always @(posedge clock)
  - b <= a;

Switch-level Modeling

- MOS switches
- nmos n1(out,data,control)
- pmos p1(out,data,control)

CMOS switch

- cmos (out, data, ncontrol, pcontrol)
- ncontrol = 1 for passing
- Equivalent to
  - nmos(out,data,ncontrol)
  - pmos(out,data,pcontrol)
- Power and Ground
  - supply1 vdd;
  - supply0 gnd;
  - assign a = vdd;
  - assign b = gnd;

Synthesis of Verilog HDL Constructs

Synthesizable Verilog Constructs

Construct Type	Keyword or Description	Notes
ports	Input, inout, output	
parameters	parameter	
Module definition	module	
Signals and variables	wire, reg, tri	Vectors are allowed

Synthesizable Verilog Constructs

Construct Type	Keyword or Description	Notes
instantiation	Module instances Primitive gate instances	Eg. my_mux m1(out,i0,i1,s);
Functions and tasks	function, task	Timing constructs ignored
procedural	always, if, then, else, case, casex, casez	initial is not supported

Synthesizable Verilog Constructs

Construct Type	Keyword or Description	Notes
Procedural blocks	begin, end, named blocks, disable	Disabling of named blocks not allowed
Data flow	assign	Delay information is ignored
loops	for, while, forever	while and forever loops must contain @(posedge clk) or @(negedge clk)

Synthesizable Verilog Operators

- All arithmetic, logical, relational, equality (except ===, !==), bit-wise, reduction, shift, concatenation and conditional are synthesizable.

Simple Examples

```
module m (out, a, b, c );
input a, b, c;
output out;
   assign out = (a & b) | c;
endmodule
```



Simple Examples

```
// If all are 2-bit vectors


module m (out , a, b, c);
input [1:0] a, b, c;
output [1:0] out;
 assign out = (a & b) | c;
endmodule
```



Simple Examples

- assign {c_out,sum} = a + b + c_in;
  // yields a Full-adder

- assign out = (s) ? i1 : i0;
  // yields a 2-to-1 multiplexer. Always the '?' construct yields a multiplexer.

- if (s)
    out = i1;
  else
  out = i0; // has the same effect

Simple Examples

- case (s)

  1’b0: out = i0;

  1’b1: out = i1;

  endcase //yields the same result(2:1 Mux)
- Large ‘case’ statements may be used to infer large multiplexers

Simple Examples

- The *for* loops may be used to form cascaded combinational logic

```
c = c_in;
for (i=0; i <= 7; i = i + 1)
        {c, sum[i]} = a[i] + b[i] + c;
c_out = c;
```

- //8-bit ripple adder

Simple Examples

- *always* statement infers both sequential and combinational circuits.
- always @(posedge clk)

  q = d; //This is a flipflop

- always @(clk or d)

  if (clk)

  q = d; // This is a latch

Simple Examples

- always @(a or b or c_in)

  {c_out, sum} = a + b + c_in;

  //This yields a full adder
- The *function* statement synthesize to combinational blocks with one output variable. The output may be a scalar or a vector.

Logic Value System

- 0 < -- > logic-0
- 1 < -- > logic-1
- z < -- > high-impedance
- x < -- > don't-care (when assigned to a variable)
- z < -- > don't-care **(in casex and casez statement)**
- x < -- > unknown (when not assigned to a variable)

Data types

- Net data type
  - wire, wor, wand, tri, supply0, supply1
  - Bit-width is specified explicitly, else the default size is one bit.

```
module wireexample (out, a, b, c, d);
input a,b,c,d;
output out;
assign out = a & b;
assign out = c | d;
endmodule
```



The Modules used

- module AND2 (A, B, Z);

```
 input A,B;
 output Z;
 // Z = A & B;
endmodule
```

- module OR2 (A, B, Z);

```
 input A,B;
 output Z;
 // Z = A | B;
 endmodule
```

Data Types

- Intermediate Translation into Generic Library modules

```
module wireexample (out, a, b, c, d);
input a,b,c,d;
output out;
AND2(a,b,out);
OR2(a,b,out);
endmodule
```



Example

- module UsesGates (BpW, BpR, Error, Wait, Clear);

```
input Error, Wait, Clear;
output BpW, BpR;
wor BpW;
wand BpR;
        assign BpW = Error & Wait;
        assign BpW = Valid | Clear;
        assign BpR = Error ^ Valid;
        assign BpR = ! Clear;
endmodule
```

Synthesized netlist



Constants

- 30 Signed, 32 bits
- -2 Signed, 32 bits
- 2'b10 Unsigned, 2 bits
- -6'd4 Unsigned, 6 bits, 2's complement of 4
- -'d10 Unsigned, 32 bits, 2's complement of 10

Synthesis of Parameters

- Parameter is a named constant
- Its size is the same as the size of the constant.
- parameter RED = -1; //32 bit, signed
- parameter READY = 2'b01;//2 bit, unsigned

Value Holders

- Value holders in Hardware:
  - Wire
  - Flip-flop
  - Latch
- Variable of net type maps onto wires
- Variable of reg type maps onto wires or registers (collection of flip-flops or latches).

Synthesis of ‘reg’ variables

```
wire Acr, Bar, Fra;
reg Trq, Sqp;
always @(Bar or Acr or Fra)
begin
    Trq = Bar & Acr;
    Sqp = Trq | Fra;
end
```

- Now, the ‘reg’ variables DO NOT synthesize as sequential memory elements.


Bat
Act
Fta
A
B
Z
A
B
Z
OR2
Sqp
Tiq

Synthesis of ‘reg’ variables

```
wire Acr, Bar, Fra;
reg Trq, Sgp;
always @(Bar or Acr or Fra)
begin
    Sqp = Trq | Fra;
    Trq = Bar & Acr;
end
```

- Now, the Trq ‘reg’ variable is used before it is assigned and hence it should be a memory element.

Synthesis of ‘reg’ variables

```
wire Sat, Ant;
 reg Fox,  Sout;
 always @ (Sat or Ant)
 begin
     if (! Sat)
              Fox = Ant;
      Sout = !Fox;
 end
```

- The variable Fox is not assigned in the ‘else’ branch and hence it is inferred as a latch. Fox should retain its old value when Sat is true.


Ant
D
Q
Sout
FD1S1A
INRB
Sat
CLK
INRB
QN

Flip-flop Vs Latch

- Understand why Fox is inferred as a latch and not a flip-flop.
- This depends on the context under which a variable is assigned a value.

Verilog Constructs to Gates

- Continuous Assignment
  - This represents in hardware, logic that is derived from the expression on the right-hand-side of the assignment statement driving the net that appears on the left-hand-side of the assignment statement.
- module Continuous (Statin, StatOut);

```
input Statin;
output Statout;
    assign Statout = ~Statin;
endmodule
```



Non-Blocking Procedural Assignment

- module NonBlocking (RegA, Mask, RegB);

```
input [3:0] RegA, Mask;
output [3:0] RegB;
reg [3:0] RegB;
always @(RegA or Mask)
        RegB <= RegA & Mask;
endmodule
```

Synthesized Circuit



Target of Assignment

- Target of a procedural assignment is synthesized to a wire, a flip-flop, or a latch, depending on the context.
- If the assignment is under the control of an edge of a clock, it infers a flip-flop.
-

```
module Target(Clk, RegA, RegB, Mask);
input Clk;
input [3:0] RegA, Mask;
output [3:0] RegB;
reg [3:0] RegB;
always @(posedge Clk)
        RegB <= RegA & Mask;
endmodule
```

Synthesized Circuit



Constant Index

- module ConstantIndex (A, C, Reg_File, Zcat);

```
input [3:0] A,C;
input [3:0] Reg_File;
output [3:0] Zcat;
        assign Zcat[3:1] = {A[2],C[3:2]};
        assign Zcat[0] = Reg_File[3];
endmodule
```


A2
A Z
INRB
A Z
INRB
ZCat3
C3
A Z
INRB
A Z
INRB
ZCat2
Reg_File3
A Z
INRB
A Z
INRB
ZCatØ
C2
A Z
INRB
A Z
INRB
ZCat1

Non-Constant Index in Expression

```
module NonComputeRight (Data, Index, Dout);
  input [0:3] Data;
  input [1:2] Index;
  output Dout;
      assign Dout = Data[Index];
endmodule
```

Synthesized circuit



Conditional Expression

```
module ConditionalExpression (StartXM, ShiftVal, Reset, StopXM);
input StartXM, ShiftVal, Reset;
output StopXM;
    assign StopXM = (! Reset)? StartXM ^ ShiftVal: StartXM | ShiftVal;
endmodule
```

- A 2-to-1 Multiplexer, with Conditional expression as control, first expression as ‘1’ option and second expression as ‘0’ option


Reset
A Z
INRB
A Z
INRB
A
B
Z
AND2
A
B
Z
OR2
StopXM
StartXM
A
B
Z
XOR2
A
B
Z
AND2
ShiftVal
A
B
Z
OR2

Always Statement

```
module EvenParity (A, B, C, D, Z);
input A, B, C, D;
output Z;
reg Z, Temp1, Temp2;
        always @(A  or B or C or D)
        begin
                Temp1 = A ^ B;
                Temp2 = C ^ D;
                Z = Temp1 ^ Temp2;
        end
endmodule
```


A
B
A
B
Z
XOR2
A
B
Z
XOR2
Z
C
D
A
B
Z
XOR2

Temporary Variables

- Each assignment (even to same variable) implies a unique wire

```
module VariablesAreTemporaries (A, B, C, D, Z);
 input A, B, C, D;
 output Z;
 reg Z;
      always @(A or B or C or D)
      begin: VAR_LABEL
                  integer T1, T2;
                  T1 = A & B;
                  T2 = C & D;
                  T1 = T1 | T2;
                  Z = ~T1;
    end
endmodule
```



If statement and Latches

```
module Increment(Phy, Ones, Z);
input Phy;
input [0:1] Ones;
output [0:1] Z;
reg [0:1] Z;
        always @(Phy or Ones)
                if (Phy) Z = Ones + 1;
endmodule      // latch inferred
```

Latch Inferred



```
module Compute (Marks, Grade);
input [1:4] Marks;
output [0:1] Grade;
reg [0:1] Grade;
parameter FAIL = 1; PASS =2; EXCELLENT = 3;
   always @(Marks)
         if (Marks < 5) Grade = FAIL;
                  else if ((Marks >= 5) & (Marks < 10))
                           Grade = PASS;
endmodule   // Again a Latch, what if Marks < 16?
```


Marks4
Marks3
Marks2
Marks1
A1
A2
B
Z
OAI21
'b1
A
B
Z
AND2
D
Q
CK
CD
QN
Grade1
FD1S1D
A Z
INRBH
A1
A2
A3
B1
B2
Z
OAI32
Grade0
FD1S1D

If statement and Latches

```
module Compute (Marks, Grade);
input [1:4] Marks;
output [0:1] Grade;
reg [0:1] Grade;
parameter FAIL = 1; PASS =2; EXCELLENT = 3;
  always @(Marks)
      if (Marks < 5) Grade = FAIL;
            else if ((Marks >= 5) & (Marks < 10))
                  Grade = PASS;
            else Grade = EXCELLENT;
endmodule
```

No latches


Marks1
Marks3
Marks2
Marks4
A Z
INRBH
A1
A2
B
Z
OAI21
A1
A2
B
Z
OAI21
A
B
Z
ND2
A
B
Z
ND2
Grade0
Grade1

Case Statement

```
module ALU (Op, A, B, Z);
input [1:2] Op;
input [0:1] A, B;
output [0:1] Z;
reg [0:1] Z;

parameter ADD = 'b00, SUB = 'b01, MUL = 'b10, DIV = 'b11;

        always @(Op or A or B)
        case (Op)
                ADD: Z = A + B;
                SUB: Z = A – B;
                MUL: Z = A * B;
                DIV: Z = A/B;
        endcase
endmodule
```


M1
A0
A1
B0
B1
Z0
Z1
MULTP
S1
SUBTR
D1
DIVDR
ADDER
INRBH
Op2
Op1
S927
AOI22
OAI22
Z0
Z1

Synthesis of casex

```
module PriorityEncoder (Select, BitPosition);
input [5:1] Select;
output [2:0] BitPosition;
reg [2:0] BitPosition;
   always @(Select)
         casex (Select)
                  5'bxxxx1 : BitPosition = 1;
                  5'bxxx1x : BitPosition = 2;
                  5'bxx1xx : BitPosition = 3;
                  5'bx1xxx : BitPosition = 4;
                  5'b1xxxx : BitPosition = 5;
                  default : BitPosition = 0;
         endcase
endmodule
```

The Semantics

- if (Select[1]) BitPosition = 1; else
  if (Select[2]) BitPosition = 2; else
  if (Select[3]) BitPosition = 3; else
  if (Select[4]) BitPosition = 4; else
  if (Select[5]) BitPosition = 5; else
  BitPosition = 0;

The Synthesized Netlist



A priority encoder using casex statement.

Inferring Latches in case statements

- If a variable is assigned a value only in some branches of a ‘case’ statement, and not in all possible branches, then, a latch is inferred for that variable.
- The rules apply equally for casex and casez statements

Example

```
module StateUpdate (CurrentState, Zip);
 input [0:1] CurrentState;
 output [0:1] Zip;
 reg [0:1] Zip;
 parameter S0 = 0, S1 = 1, S2 = 2, S3 = 3;
 always @(CurrentState)
    case (CurrentState)
    S0, S3: Zip = 0;
    S1: Zip = 3;
    endcase
  endmodule
```



Latch inferred for a variable in a case statement.

To Avoid Latches

```
module StateUpdate (CurrentState, Zip);
  input [0:1] CurrentState;
  output [0:1] Zip;
  reg [0:1] Zip;
  parameter S0 = 0, S1 = 1, S2 = 2, S3 = 3;
  always @(CurrentState)
  begin
    Zip = 0;    //This statement is added – mimics 'default'
          case (CurrentState)
                  S0, S3: Zip = 0;
                  S1: Zip = 3;
          endcase
  end
   endmodule
```

Synthesis of For Loops

- Unrolling the For loop – All statements within the For loop are replicated, once for each value of the For-loop index.
- The restriction is that the loop bounds have to be constants – else synthesis is beyond any scope.

Example

- module Demultiplexer(Address, Line);
- input [1:0] Address;
- output [3:0] Line;
- reg [3:0] Line;
- integer J;
- always @(Address)
- for (J = 3; J >= 0; J = J –1)
- if (Address == J) Line[J] = 1;
- else Line[J] = 0;
- endmodule

Loop Unrolling

```
if (Address == 3) Line[3] = 1; else Line[3] = 0;
if (Address == 2) Line[2] = 1; else Line[2] = 0;
if (Address == 1) Line[1] = 1; else Line[1] = 0;
if (Address == 0) Line[0] = 1; else Line[0] = 0;
```

The Synthesized Netlist



Multi-phase Clocks

```
module MultiPhaseClocks (Clk, A,B,C,E);
 input Clk, A, B, C;
output E;
reg E,D;
always @(posedge Clk)
  E <= D | C;
always @(negedge Clk)
  D <= A & B;
endmodule
```

The Synthesized Circuit



References

- Verilog HDL (2nd Edition)
  Samir Palnitkar, Prentice Hall Publications, 2003
  ISBN : 0130449113
- J. Bhasker's, "Verilog HDL Synthesis – A Practical Primer" – book.
- HDL Chip Design : A Practical Guide for Designing, Synthesizing and Simulating ASICs and FPGAs Using VHDL or Verilog
  Douglas J Smith, Doone Publications, 1998
  ISBN : 0965193438